

第2章

FRAME #/IRDY #を読みDEVSEL #/TRDY #を操る

PCIバス・トランザクション
の基礎

来須川 智久

PCIバスに準拠するためにはコンフィグレーション・レジスタが不可欠だが、PCIバスを理解することを優先し、ここではコンフィグレーション空間をもたない、アドレスを固定してデコードすることでメモリ・サイクルに回答するPCIデバイスを設計する。第1章で説明したように、基本的に各トランザクションはすべて同じタイミングで制御されている。この共通した基本ルールを明確にし、これを理解することでPCIバスをマスターできる。

(編集部)

第2部

メモリ・ライト・サイクル — ターゲット1の設計

● トランザクションの本質を理解するために

いきなりPCIバスに完全準拠したデバイスを設計すると、コンフィグレーション空間や基本的なトランザクションに付随する細かなタイミングを考慮することが必要で、あれこれ説明しているうちにPCIバス・トランザクションの本質を見逃してしまうおそれがあります。

そこでまずはじめに、コンフィグレーション空間をもたずにデコード・アドレスをハードウェア的に固定(ソース・コード上で定数として定義)してしまい、しかも対応するトランザクションはメモリ・ライト・サイクルのみ(書き込み専用)という、PCIバスの規格からすれば言語道断といってよいほど仕様を削ったデバイスを設計します。

最初に設計するデバイスをターゲット1と名付け、少しずつ機能を追加していきながらターゲット2, 3と設計していきます。

なお、本書で設計した設計データを、実際にデバイスに書き込んでPCIバスで使う場合は、まずコラム1を参照していただき、そのまま使用できるかどうかを確認してください。

● VHDLソースの雛形

図1にVHDLソースの雛形を示します。以後の解説ではこのソースの各部に、各設計に必要なソースをはめ込んでいったり、機能を追加していくスタイルを採ります。




図1 PCIターゲット・デバイス設計のVHDLソースの雛形

見本




図2
メモリ・アドレス8000_0000h番地に
データを書き込む

2.1 シーケンサの骨格

● ハードウェアの仕様

まず、最初に設計するハードウェアをターゲット1と名付けました。このデバイスの仕様を次に示します。

- 1) ターゲット・デバイスのローカル側にLEDを1個接続
- 2) メモリ・ライト・トランザクションに回答するPCIターゲット・デバイス
- 3) 占有メモリ・アドレスは8000_0000h番地固定
- 4) 書き込み時のAD[0]ビットの値が“1”ならLEDを点灯/“0”なら消灯

ISAバスならあっという間にできそうな、非常に簡単なハードウェアです。このときのPCIバスの動作を

予想した波形が、図2になります。DEVSEL#とTRDY#をこの図のようなタイミングで制御すればよいわけです。

● ステート・マシン

図2のように、クロック1で をして、クロック2で がアサートされたら をして...といった制御をするには、ステート・マシンと呼ばれる回路が使われます。そのためには、クロック1で必要な仕事は何か、クロック2でどの信号が変化のを待つのか...ということを洗い出す必要があります(図3)。

これが、PCIバスを制御するステート・マシンの内容となります。これをPCIターゲット・シーケンサと名付け、VHDLではターゲット・シーケンサ・プロセスとしてprocess文を記述します。また、このほかに実際にアドレスをデコードする回路、そして

コラム1 テスト環境の確認

ターゲット1とターゲット2ではデコード・アドレスを8000_0000hのアドレスに固定しているため、この設計データを実際にデバイスに書き込んでPCIバスで使う場合、その環境ですでにほかのPCIデバイスによって8000_0000hのアドレス領域が使われていると正常に動作しません。

また、後述するコンフィグレーション・レジスタのベンダIDとデバイスIDも、場合によってはすでに同じIDが使われている可能性もあります。

そこで、PCIボードの動作確認に使うPC/AT互換機のPCI環境をチェックするPCCHK.EXEというプログラムを用意しました(本書付属CD-ROMに収録)。場合によっては、デコード・アドレスやベンダID/デバイスIDの変更が必要な場合があります。

ただし、このチェックでOKが出ても第10章のコラム3(p.148)のようなマザーボードでは、ターゲット1とターゲット2は動かない場合があります。詳細はAppendix2を参照してください。

リスト1 ターゲット1の port宣言部

```
entity PCI_TGT1 is
  port (
    -- PCIバス信号ピン(ターゲット1で使用する信号) --
    PCICLK : in std_logic; -- PCIバス・クロック
    RST_n : in std_logic; -- 非同期リセット
    PCIAD : inout std_logic_vector(31 downto 0); -- アドレス/データ・バス
    C_BE_n : in std_logic_vector(3 downto 0); -- PCIバス・コマンド/バイト・イネーブル
    FRAME_n : in std_logic; -- フレーム
    IRDY_n : in std_logic; -- イニシエータ・レディ
    DEVSEL_n : out std_logic; -- デバイス・セレクション
    TRDY_n : out std_logic; -- ターゲット・レディ
    -- LED制御ピン
    LED_OUT : out std_logic
  );
end entity PCI_TGT1;
```

リスト2 ターゲット1の レジスタ宣言/ 定数定義部と 同時処理文

```
architecture RTL of PCI_TGT1 is
  -- ***** レジスタ/定数 定義部分
  -- PCIバス・コマンド/アドレス/IDSELホールド・レジスタ --
  signal PCI_BusCommand : std_logic_vector(3 downto 0); -- PCIバス・コマンド・レジスタ
  signal PCI_Address : std_logic_vector(31 downto 0); -- PCIアドレス・レジスタ

  -- ローカル・バス・シーケンサ スタート・フラグ
  signal LOCAL_Bus_Start : std_logic;
  -- ローカル・バス・シーケンサ データ転送完了フラグ
  signal LOCAL_DTACK : std_logic;

  -- トライステート・バッファ制御用のフリップフロップ定義 --
  -- PCIバス信号線
  signal PCIAD_HiZ : std_logic; -- ADポート出力ドライブ制御
  signal PCIAD_Port : std_logic_vector(31 downto 0); -- ADポート出力レジスタ
  signal DEVSEL_HiZ, DEVSEL_Port : std_logic; -- DEVSEL#出力ドライブ制御/出力レジスタ
  signal TRDY_HiZ, TRDY_Port : std_logic; -- TRDY#出力ドライブ制御/出力レジスタ

  -- PCIバス・コマンド(ビット・パターン定義) --
  -- メモリ・リード・サイクル
  constant PCI_MemWriteCycle : std_logic_vector(3 downto 0) := ("0111");

  -- アドレス・デコード・フラグ
  signal Hit_Device : std_logic; -- デバイス・ヒット

begin
  -- ***** 同時処理文
  -- トライステート・バッファ動作
  PCIAD <= (others => 'Z') when PCIAD_HiZ = '1' else PCIAD_Port;
  DEVSEL_n <= 'Z' when DEVSEL_HiZ = '1' else DEVSEL_Port;
  TRDY_n <= 'Z' when TRDY_HiZ = '1' else TRDY_Port;
```

1でinとして宣言した信号は、後述するプロセス文の中でいつでも状態を参照可能、つまり代入文の右辺に使えます〔図5(a)〕。

LED出力は常時信号を出力するので、リスト1のようにoutと宣言するだけで、代入文で値を代入することで信号を出力します〔図5(b)〕。outで宣言したポートは出力専用なので代入文の右辺には使えません。

また、PCIターゲット・デバイスを設計する場合、DEVSEL#やTRDY#はトライステートで制御する出力信号です。port宣言ではoutとして宣言します。さらに、リスト2のsignal宣言で信号を定義し、同時処理文との組み合わせで、ハイ・インピーダンスの制御をします〔図5(c)〕。

たとえば、DEVSEL_HiZに“1”を代入すると、実際のDEVSEL#の出力にあたるDEVSEL_nには“Z”が代入され、ハイ・インピーダンスになるわけです。DEVSEL_HiZが“1”以外の値ならば、DEVSEL_Portの内容がDEVSEL_nに出力されます。DEVSEL_nに何を出力するかは同時処理文に記述するので、DEVSEL_HiZやDEVSEL_Portに値を代入した瞬間にDEVSEL_nの出力内容も変わります。

ADバスはデータの入出力があるのでinoutとして双方向で宣言します。これもトライステート制御なので、リスト2に示すようにトライステート制御用にsignal宣言で信号を定義し、同時処理文で動作を記述します〔図5(d)〕。以降は、それぞれのブロックごとに、処理内容の詳細を説明します。

2.2 PCIターゲット・シーケンサ

図3のPCIバスの制御手順をそのままシーケンサにした状態遷移図を図6に、PCIターゲット・シーケンサ・プロセス部分のVHDLのソースをリスト3に示します。また、このシーケンサを実際にPCIバスで動作させた場合のタイミングを図7に示します。

それでは、各ステートがどのような意味をもち、どのような処理をしているのかについて詳しく解説します。

● リセット処理

PCIバスにおけるリセットは、RST#がアサートされることにより行われます。これは非同期に行われるので、クロック・イベント行より先に判定しています。

リセット時には、ステート・マシンの状態を示す変数をバス・アイドル状態にセットし、アドレスやバス・コマンドを保持するレジスタや内部変数をクリアします。


また、ターゲットがドライブするDEVSEL#とTRDY#、およびADバス(ターゲット1ではドライブしないが)をハイ・インピーダンス状態に設定します。

● BUS_IDLEステート

これは、ターゲット・シーケンサがアイドル状態のときのステートです。PCIバス上でトランザクションが開始されるまで、このBUS_IDLEステートにとどまり続けます。


ここではトランザクションの開始は、FRAME#がアサートされIRDY#がデアサートされているという条件で判定しています。

ここで、ADバスの状態をアドレスとしてPCIアドレス・レジスタ(PCI_Address)に取り込み、C/BE#の状態をバス・コマンドとしてPCIバス・コマンド・レ




port宣言 FRAME_n:in std_logic;

(a) 入力ポート




port宣言 LED_OUT:out std_logic;

(b) 出力ポート



```
port宣言 DEVSEL_n:out std_logic;
signal宣言 signal DEVSEL_HiZ:std_logic;
signal宣言 signal DEVSEL_Port:std_logic;
同時処理文 DEVSEL_n<='Z'when DEVSEL_HiZ='1'
else DEVSEL_Port;
```

(c) トライステート出力ポート



```
port宣言 PCIAD:inout std_logic_vector (31 downto 0);
signal宣言 signal PCIAD_HiZ:std_logic;
signal宣言 signal PCIAD_Port:std_logic_vector (31 downto 0);
同時処理文 PCIAD<=(others=>'Z')when
PCIAD_HiZ='1'else PCIAD_Port;
```

(d) トライステート入出力ポート

図5 入力/出力/トライステート出力/トライステート入出力ポート




図6 PCIターゲット・シーケンサの状態遷移図

リスト3 PCIターゲット・シーケンサ・プロセス部分のVHDLのソース

```

-- ***** PCIターゲット・シーケンサ
PCI_TGT_Seq : process( PCICLK, RST_n )

-- PCIターゲット・シーケンサ・ステート・バリュウ・レジスタ --
variable PCI_CURRENT_STATE : std_logic_vector (2 downto 0);-- 現在のステート
variable PCI_NEXT_STATE    : std_logic_vector (2 downto 0);-- 次のステート

-- PCIターゲット・シーケンサ・ステート・マシン定義
constant BUS_IDLE          : std_logic_vector (2 downto 0) := "000";
constant ADRS_COMPARE      : std_logic_vector (2 downto 0) := "010";
constant WAIT_IRDY         : std_logic_vector (2 downto 0) := "011";
constant WAIT_LOCAL_ACK    : std_logic_vector (2 downto 0) := "100";
constant ACC_COMPLETE      : std_logic_vector (2 downto 0) := "101";
constant BUS_BUSY          : std_logic_vector (2 downto 0) := "110";
constant TURN_AROUND       : std_logic_vector (2 downto 0) := "111";

begin

-- ***** リセット時動作 ***** --
if (RST_n = '0') then
    -- PCIバス・リセット時(非同期リセット)

    PCI_CURRENT_STATE := BUS_IDLE;    -- ステート・マシン IDLE状態 リセット
    PCI_NEXT_STATE    := BUS_IDLE;    -- ステート・マシン IDLE状態 リセット

    LOCAL_Bus_Start <= '0';          -- ローカル・バス・シーケンサ スタート・フラグ クリア

    PCI_BusCommand <= (others => '0'); -- PCIバス・コマンド・レジスタ クリア
    PCI_Address <= (others => '0');    -- PCIバス・アドレス・レジスタ クリア

    -- 制御出力端子をハイ・インピーダンス
    PCIAD_HiZ <= '1';
    DEVSEL_HiZ <= '1'; DEVSEL_Port <= '1'; -- DEVSEL#="H"
    TRDY_HiZ <= '1'; TRDY_Port <= '1'; -- TRDY#="H"

-- ***** PCIターゲット・シーケンサ・ステート・マシン ***** --
elseif (PCICLK'event and PCICLK = '1') then

    PCI_CURRENT_STATE := PCI_NEXT_STATE;-- ステート・マシン制御
    case PCI_CURRENT_STATE is
        -- ***** BUS_IDLE時の動作 ***** --
        when BUS_IDLE =>-- トランザクションの開始待ち

            if (FRAME_n = '0' and IRDY_n = '1') then -- トランザクション開始
                PCI_BusCommand <= C_BE_n;          -- PCIバス・コマンド取得
                PCI_Address <= PCIAD;              -- アドレス取得
                PCI_NEXT_STATE := ADRS_COMPARE;
            else -- バス・アイドル時このステートにとどまる
                PCI_NEXT_STATE := BUS_IDLE;
            end if;

            -- ***** ADRS_COMPARE時の動作 ***** --
            when ADRS_COMPARE => -- アドレス・デコード結果を調べる

                if (Hit_Device = '1') then -- 自分が選択された
                    DEVSEL_Port <= '0'; DEVSEL_HiZ <= '0'; -- DEVSEL#アサート
                    TRDY_HiZ <= '0'; -- TRDY# を"H"にドライブ
                    PCI_NEXT_STATE := WAIT_IRDY; -- イニシエータ・レディを待つステートへ
                else -- 自分が選択されていない

                    PCI_NEXT_STATE := BUS_BUSY;
                    -- トランザクションの終了を待つステートへ
                end if;

            -- ***** BUS_BUSY時の動作 ***** --
            when BUS_BUSY =>-- トランザクション終了待ち

                if (FRAME_n = '1' and IRDY_n = '1') then -- トランザクション終了(アイドル)
                    PCI_NEXT_STATE := BUS_IDLE; -- トランザクション開始待ちステートへ
                else -- トランザクション中ならこのステートにとどまる
                    PCI_NEXT_STATE := BUS_BUSY;
                end if;

            -- ***** WAIT_IRDY時の動作 ***** --
            when WAIT_IRDY => -- イニシエータ・レディ待ち

                if (IRDY_n = '0') then-- イニシエータの準備完了
                    LOCAL_Bus_Start <= '1';-- ローカル・バス・シーケンサ スタート!
                    PCI_NEXT_STATE := WAIT_LOCAL_ACK;-- ローカル・バス・シーケンサ終了待ちステートへ
                else -- イニシエータの準備がまだならこのステートにとどまる
                    PCI_NEXT_STATE := WAIT_IRDY;
                end if;
    end case;
end if;

```

リセット時
の処理

クロックの立ち上がりに同
期して動作するステート・
マシンの記述

BUS_IDLEステート

ADRS_COMPAREステート

BUS_BUSYステート

WAIT_IRDY
ステート

リスト3 PCIターゲット・シーケンサ・プロセス部分のVHDLのソース(つづき)

```

-- ***** WAIT_LOCAL_ACK時の動作 ***** --
when WAIT_LOCAL_ACK =>
  -- ローカル・バス・シーケンサ終了待ち

  LOCAL_Bus_Start <= '0';-- ローカル・バス・シーケンサ スタート・フラグ クリア
  if (LOCAL_DTACK = '1') then -- ローカル・バス・シーケンサ データ転送完了
    TRDY_Port <= '0';-- TRDY# アサート
    PCI_NEXT_STATE := ACC_COMPLETE;-- アクセス完了スタートへ
  else -- ローカル・バス・シーケンサの準備がまだならこのスタートにとどまる
    PCI_NEXT_STATE := WAIT_LOCAL_ACK;
  end if;
} WAIT_LOCAL_ACK
ステート

-- ***** ACC_COMPLETE時の動作 ***** --
when ACC_COMPLETE => -- アクセス完了スタート

  DEVSEL_Port <= '1';-- DEVSEL# ディアサート
  TRDY_Port <= '1';-- TRDY# ディアサート
  PCIAD_HiZ <= '1'; -- PCIAD[31:0]バス・ドライブ解放
  PCI_NEXT_STATE := TURN_AROUND;-- ターンアラウンド・スタートへ
} ACC_COMPLETEステート

-- ***** TURN_AROUND時の動作 ***** --
when TURN_AROUND => -- ターンアラウンド・スタート

  DEVSEL_HiZ <= '1'; -- DEVSEL#ドライブ解放
  TRDY_HiZ <= '1'; -- TRDY#ドライブ解放
  PCI_NEXT_STATE := BUS_IDLE;-- トランザクション開始待ちスタートへ
} TURN_AROUNDステート

-- *****
when others => null; -- これ以外の値では何もしない場合でも必ず入れる

end case;

end if;

end process PCI_TGT_Seq;

```

PCIターゲット・シーケンサ・ステート番号(図6)




図7 ターゲット1の動作

レジスタ(PCI_BusCommand)に取り込みます。アドレス・フェーズは、バス・アイドル状態からFRAME#がアサートされた最初のPCIクロックの立ち上がりのときだけです。アドレスとバス・コマンドをレジスタに取り込んだら、ADRS_COMPAREステートへ移行します。

ここではFRAME#がアサートされたという条件以外に、さらにIRDY#がディアサートされていることもアドレス・フェーズの認識条件に追加し、より厳密にアドレス・フェーズを認識しています。FRAME#がアサートされた条件だけでは、FRAME#とIRDY#が同時にアサートされたときもトランザクションが開始されたと判定してしまいます。

FRAME#とIRDY#が同時にアサートされることは、PCIバスの規格上認められていません。もしこのようなアクセスが発生したとしたらイニシエータ・デバイスの動作不良であるとみなし、DEVSEL 応答を返さないようにしています。

もっとも、最悪の条件を考えすぎてもあまり意味がないので、現状ではFRAME#がアサートされた条件だけでトランザクションの開始を判定しても、実質的には問題ないでしょう。

トランザクションの開始が判定できなければ、このステートにとどまります。ここでは、同じステートにとどまる場合は、次のステートの状態を保持するPCI_NEXT_STATEに同じステートの値を代入しています。VHDLの記述では代入してもしなくても、すでにその値が入っているわけなので、このelse文は省略できますが、条件が成立していなければ同じステートにとどまることを明示的に示す意味で記述しています。ほかのステートでも同様です。

● ADRS_COMPAREステート

アドレス・デコード結果を調べて、現在発生しているトランザクションが自分にあてられたものか、またはほかのターゲットのものかを判定します。Hit_Deviceが“1”なら、自分が選択されていることになり、DEVSEL#をアサートしてイニシエータにDEVSEL 応答を返し、WAIT_IRDYステートへ移行します。

また同時に、TRDY#のドライブも開始します。TRDY#の信号レベルはリセット時に“1”を入れているので、“H”レベルとなります。

PCIバス規格ではこのDEVSEL 応答を、FRAME#がアサートされたことを認識してから3クロックまでの間に行わなければなりません。速さによって高速/

中速/低速応答と呼ばれますが、今回の設計では中速応答になります。

もし、Hit_Deviceが“1”ではない場合、ほかのターゲットへのアクセスであると判断して、BUS_BUSYステートへ移行します。

● BUS_BUSYステート

BUS_BUSYステートは、ほかのターゲットに対するトランザクションが発生しているとき、そのトランザクションが終了するまで待避するステートです。

たとえば、直前のADRS_COMPAREステートで、発生したトランザクションが自分宛てのものではないと判定したあと、そのままBUS_IDLEステートに戻ってもよさそうなものですが、それはできません。

PCIバス・トランザクションは一つのアドレス・フェーズと、それに続く一つもしくは複数のデータ・フェーズにより成り立っています。一つのトランザクション中に複数のデータ・フェーズが存在する転送を、バースト転送と呼んでいます。図8がバースト転送時のPCIバスのタイミング例です。この図からわかるように、アドレス・フェーズから最後のデータ・フェーズの直前のデータ転送まで、FRAME#がアサートされ続けます。

もう一つ注意が必要な場合があります。イニシエータが出力するFRAME#とIRDY#のアサート/ディアサート・タイミング波形のいくつかを図9に示します。一見してわかるように、図9(b)セットでは、バースト転送でもないのにFRAME#が数クロックの間アサートされることがあるのです。

よって、ADRS_COMPAREステートからそのままBUS_IDLEステートにもどってしまうと、バースト転送時や単一データ・フェーズのトランザクションでも、FRAME#が数クロックの間アサートされるようなイニシエータの場合、FRAME#がアサートされている途中をアドレス・フェーズだと誤って認識し、再度ADRS_COMPAREステートに移行してしまいます。

そこで、自分に対するアクセスではなかった場合は、そのトランザクションが終了するまで待つ必要があります。

PCIバスのトランザクションが終わったかどうかはバス・アイドル状態、すなわちFRAME#とIRDY#がともにディアサートになったことで判定します。

このステートでは、これを検出するまで何らかのトランザクションがほかのエージェントによって行われていると判断してこのステートにとどまり、トランザ






図8
バースト転送時のPCIバスの動き



(a) アドレス・フェーズでFRAME が1クロックだけ“L”



(b) アドレス・フェーズ後もFRAME が数クロック“L”

図9
トランザクション開始時の
FRAME#の動き

クションが終了したら BUS_IDLE ステートに移行します。

● WAIT_IRDYステート

WAIT_IRDYステートは、イニシエータ側のデータ

転送準備が整ったことを示す IRDY# 信号のアサートを待ち、アサートされたらローカル・バス・シーケンサを起動させるステートです。

図7を見ると、DEVSEL 応答を返す段階で、すで

に IRDY# がアサートされている(クロック 4)から、あえてこの状態で明示的に調べる必要はないように思えます。しかし、図 9 b) のような、アドレス・フェーズから数クロック後にイニシエータ側のデータ送受信準備が整うデバイスが存在します。

図 9 b) の波形とここで設計したシーケンサを照らし合わせると、この WAIT_IRDY ステートは、図 9 a) のクロック 4 に位置します。つまり、図 9 a) のタイミングではほとんどの場合、このステートにくる段階ではすでに IRDY# がアサートされているのに比較し、図 9 b) のタイミングではまだ IRDY# がアサートされていない可能性があるわけです。

よって、WAIT_IRDY ステートでは、IRDY# 信号がアサートされるまで、つまりイニシエータ側のデータの転送準備が整ったということが通知されるまではこのステートにとどまります。

IRDY# がアサートされたら、次はいよいよ LED の点灯制御処理が必要です。LED 点灯制御回路は別のブロックで定義しているので、このターゲット・シーケンサで行うことは、その別ブロックに対して「動き出してもいいよ」という起動許可のフラグをセットすることです。

それが、LOCAL_Bus_Start という名前で宣言されたフラグ・レジスタに“1”をセットすることです。これで LED 点灯制御回路、つまりローカル・バス・シーケンサが動き出します。

LOCAL_Bus_Start フラグをセットしたら、次は WAIT_LOCAL_ACK ステートに移行します。

● WAIT_LOCAL_ACK ステート

LED 点灯制御回路の動作が完了したかどうか、つまりローカル・バス・シーケンサのデータ転送が終了したかどうかを待ちうけるステートです。

すでにローカル・バス・シーケンサはスタートしたので、直前の WAIT_IRDY ステートでセットした LOCAL_Bus_Start フラグをクリアし、ローカル・バス・シーケンサのデータ転送完了フラグ LOCAL_DTACK が“1”になるのを待ちます。

このフラグがセットされるまで、何もせずにこのステートにとどまり続けているので、その間イニシエータは IRDY# をアサートし続けたまま、ターゲットからのデータ転送終了を示す TRDY# 信号がアサートされるのを待っています。つまり、データ・フェーズが継続されています。

そして、LOCAL_DTACK フラグが“1”にセットされ

たら、データ・フェーズを正常に終了したことをイニシエータに通知するために TRDY# 信号をアサートし、ACC_COMPLETE ステートに移行します。

● ACC_COMPLETE ステート

直前の WAIT_LOCAL_ACK ステートで TRDY# をアサートしましたが、その瞬間にイニシエータがデータ転送の完了を認識するわけではありません。

PCI バスはクロック同期ですから、実際にはこの ACC_COMPLETE ステートのクロックで、イニシエータは TRDY# がアサートされたことを認識します。

イニシエータが正しく TRDY# を認識しているはずなので、ターゲットも TRDY# と DEVSEL# をディアサートしてデータ転送を完了します。

ここが同期回路設計の初心者、また PCI バスの初心者が不安に思う点の一つでしょう。ほんとうにイニシエータはここで TRDY# を認識してくれるの？ IRDY# がディアサートされるのを確認してから TRDY# をディアサートしたほうが安心だ……気持ちわかりますが、それでは同期回路設計ではありません。安心してここでディアサートしてください。

さて、データ転送が終わったからといって、すぐに BUS_IDLE ステートに戻ることはできません。“H”にした次のクロックで信号のドライブを切り離し、ハイ・インピーダンスにする処理、サステインド・トライステートの処理がまだ終わっていません。トランザクションを完全に終了するために、次に TURN_AROUND ステートに移行します。

● TURN_AROUND ステート

この TURN_AROUND ステートでは、直前の ACC_COMPLETE ステートでディアサートした DEVSEL# と TRDY# のドライブを切り離し、ハイ・インピーダンス状態にします。そして、BUS_IDLE ステートへ移行します。ACC_COMPLETE ステートでディアサート (“H”)し、このステートでハイ・インピーダンスにすることで、サステインド・トライステートが実現できます。

ステート・マシンを知っている読者の中には、たんにディアサートした信号線をハイ・インピーダンスにするなら、ACC_COMPLETE ステートから直接 BUS_IDLE ステートへ移行して、BUS_IDLE ステートで行ってもよいのではないと思われるでしょう。たしかに、そうやってステート数を削ることは可能です。

しかしここでは、自分の出力する DEVSEL# や TRDY# 信号をハイ・インピーダンス状態にするス




図10
トランザクションが
連続して発生した場合

テートを用意することで、トランザクションの終了処理を明確に定義するとともに、BUS_IDLE状態は、あくまでアドレス・フェーズの開始を待つ処理だけをさせることで、それぞれの状態の本来の目的に専念できるわけです。はじめてPCIバスを理解しようとする場合には、これはたいへん重要なポイントだと思います。

また、TURN_AROUND状態を設けたことにより、連続して発生するトランザクションを正しく認識できるのかどうか心配になります。図10に、あるトランザクションの後半と、それに続くトランザクションが連続して発生したときのタイミングを示します。

- 1) WAIT_LOCAL_ACK状態でローカル・バス側の処理の終了通知を受けるとTRDY#をアサートし、ACC_COMPLETE状態に移行する(クロック1)
- 2) クロック2でTRDY#をディアサートし、TURN_AROUND状態に移行する
- 3) クロック3でDEVSEL#とTRDY#をハイ・インピーダンス状態にする。これでこのトランザクションのすべての処理を終了し、BUS_IDLE状態へ戻る。トランザクションが連続する場合はこのクロックでFRAME#がアサートされ、アドレス・フェーズが始まる
- 4) クロック4ですでにFRAME#がアサートされているため、トランザクションが開始されたと判断し、ADバスとC/BE#信号の値を内部レジスタに保持し、ADRS_COMPARE状態に移行する
- 5) 以降はこれまでの説明と同様に、アドレス・デコードの結果によりBUS_BUSY状態またはWAIT_IRDY状態に移行する

このようにTURN_AROUND状態を導入しても、連続するトランザクションを正しく認識することが可能です。

2.3 アドレス・デコーダ

ターゲット・シーケンサの説明では、BUS_IDLE状態においてアドレスとバス・コマンドを取り込み、次のADRS_COMPARE状態でデコード結果のHit_Deviceを見て判定する、と説明しました。このとき、この状態の間でひそかにアドレス・デコードを行っているのが、このアドレス・デコード・プロセスです。リスト4にアドレス・デコード・プロセスのVHDLソースを示します。

アドレス・デコードでは、PCI_AddressとPCI_BusCommandをデコードして、ターゲット1の仕様である、

- アドレスが8000_0000h番地が
 - バス・コマンドがメモリ・ライト・コマンドか
- の二つの条件が成立するかどうかをつねに調べています。このソースは、回路的には図11のようになります。つまり、ターゲット・シーケンサのBUS_IDLE状態でアドレスとバス・コマンドを取り込むと、その値がそのままこのアドレス・デコード回路に入力されます。回路図を見てわかるように、クロック同期でも順序回路でもない組み合わせ回路です。ゲート遅延時間後にはデコード結果であるHit_Device信号が出力されます。

VHDLのソース・リストだけを見ていると、ハードウェアというよりはソフトウェアというイメージが

リスト4
アドレス・デコード・プロセスの
VHDLソース

```

-- ***** アドレス・デコード・シーケンサ
Address_Compare : process (
    PCI_Address,    -- PCIバス・アドレス
    PCI_BusCommand -- バス・コマンド
)
    constant MEMORY_BASE_ADDRESS : std_logic_vector(31 downto 0) := X"80000000"; -- デコード・アドレス
begin
    -- メモリ空間へのアクセス・アドレスとベース・アドレスが一致したか
    if (
        PCI_BusCommand(3 downto 0) = PCI_MemWriteCycle-- メモリ・ライト・サイクル
    ) and (
        PCI_Address = MEMORY_BASE_ADDRESS-- ベース・アドレス比較
    )
    then
        Hit_Device <= '1';    -- メモリ・サイクル・ヒット
    else
        Hit_Device <= '0';
    end if;
end process Address_Compare ;
    
```




図11
アドレス・デコード・プロセスの回路

先行し、同時に動いている、つねに動いている、という部分が直感的にわかりづらいのではないかと思います。とくに、ソフトウェア技術者にはその傾向が強いのではないのでしょうか。

2.4 ローカル・バス・シーケンサ

● LED点灯制御回路

ターゲット1の仕様では、本当に書き込みがうまく行っているかどうかを確認するためにLEDを制御します。このLEDはカソード側をLED_OUTピンに直結し、LEDのアノード側は1k程度の抵抗を間にはさんでV_{cc}電源に接続します。こうすることにより、LED_OUTピンが“L”になればLEDが点灯するという

わけです。

● シーケンサの動作

ローカル・バス・シーケンサ、つまりLEDの点灯制御を行うシーケンサです。このシーケンサもPCIバス・クロックに同期して動作させています。図12にローカル・バス・シーケンサの状態遷移図を、リスト5にローカル・バス・シーケンサのVHDLソースを示します。では、各ステートがどのような意味をもち、どのような処理をしているかを詳しく解説します。

● リセット処理

ローカル・バスにもリセット処理は必要です。リセット信号はPCIバスのリセット信号を使いました。これもクロック・イベント行より先に記述します。

リセット時には、ステート・マシンの状態を示す変




図12
ローカル・バス・シーケンサの状態遷移図

リスト5 ローカル・バス・シーケンサのVHDLソース

```

-- ***** ローカル・バス・シーケンサ
LOCAL_BUS_Seq : process( PCICLK, RST_n )

-- ローカル・バス・シーケンサ ステートバリュウ・レジスタ --
variable LOCAL_CURRENT_STATE : std_logic_vector (1 downto 0);-- 現在のステート
variable LOCAL_NEXT_STATE    : std_logic_vector (1 downto 0);-- 次のステート

-- ローカル・バス・シーケンサ ステート・マシン定義
constant LOCAL_IDLE          : std_logic_vector(1 downto 0) := "00";
constant LOCAL_MEM_ACCESS    : std_logic_vector(1 downto 0) := "01";
constant LOCAL_STATE_COMP    : std_logic_vector(1 downto 0) := "11";

begin

-- ***** リセット時動作 ***** --
if ( RST_n = '0' ) then-- PCIバス・リセットがアサートされたとき
    -- ステートバリュウ・レジスタ クリア
    LOCAL_CURRENT_STATE := LOCAL_IDLE;    -- ローカル・バス・シーケンサ リセット
    LOCAL_NEXT_STATE    := LOCAL_IDLE;    -- ローカル・バス・シーケンサ リセット
    LOCAL_DTACK <= '0'; -- ローカル・バス・シーケンサ データ転送完了フラグ クリア
    LED_OUT <= '1'; -- LED消灯
}

-- ***** ローカル・バス・シーケンサ ステート・マシン ***** --
elsif (PCICLK'event and PCICLK = '1') then
    LOCAL_CURRENT_STATE := LOCAL_NEXT_STATE ;
    case LOCAL_CURRENT_STATE is
        -- ***** LOCAL_IDLE時の動作 ***** --
        when LOCAL_IDLE =>-- ローカル・バス・シーケンサ スタート指示待ち
            if (LOCAL_Bus_Start = '1' ) then -- ローカル・バス・シーケンサ スタート!
                LOCAL_NEXT_STATE := LOCAL_MEM_ACCESS;-- メモリ・アクセス・ステートへ
            else-- ローカル・バス・シーケンサ スタート・フラグがまだならこのステートにとどまる
                LOCAL_NEXT_STATE := LOCAL_IDLE;
            end if;
        }

        -- ***** LOCAL_MEM_ACCESS時の動作 ***** --
        when LOCAL_MEM_ACCESS => -- メモリ・サイクル
            LED_OUT <= not PCIAD(0); -- LED点灯制御
            LOCAL_DTACK <= '1' ;-- ローカル・バス・シーケンサ データ転送完了フラグ セット
            LOCAL_NEXT_STATE := LOCAL_STATE_COMP;
        }

        -- ***** LOCAL_STATE_COMP時の動作 ***** --
        when LOCAL_STATE_COMP => -- ローカル・バス・アクセス完了
            LOCAL_DTACK <= '0' ;-- ローカル・バス・シーケンサ データ転送完了フラグ セット
            LOCAL_NEXT_STATE := LOCAL_IDLE;
        }

        -- *****
        when others => null; -- これ以外の値では何もしない場合でも必ず入れる

    end case;

end if;

end process LOCAL_BUS_Seq ;

```

数をローカル・バス・アイドル状態にセットし、内部変数をクリアします。また、リセットということでLEDを消灯します。

● LOCAL_IDLEステート

ここでは、LOCAL_Bus_Start フラグに“1”がセットされるまで待ち、セットされたら LOCAL_MEM_ACCESSステートへ移行します。

● LOCAL_MEM_ACCESSステート

ここでは実際に、LEDの点灯を制御します。仕様で

は最下位ビットであるAD0が“1”なら点灯，“0”なら消灯なので、LEDを実際に点灯させるLED_OUTピンの論理内容とちょうど逆であることがわかります。

そこで、AD[0]の内容を反転してLED_OUT信号に出力します。そして、必要なデータ転送が終了したので、LOCAL_DTACKフラグに“1”をセットして、LOCAL_STATE_COMPステートへ移行します。

● LOCAL_STATE_COMPステート

ターゲット・シーケンサはLOCAL_DTACKフラグを

